Thermochimica Acta, 80 (1984) 185–189
Elsevier Science Publishers B.V., Amsterdam – Printed in The Netherlands

Note

COMPUTER-DETERMINED KINETIC PARAMETERS FROM TG CURVES. PART X

LEO REICH and S.S. STIVALA

Department of Chemistry and Chemical Engineering, Stevens Institute of Technology, Hoboken, NJ 07030 (U.S.A.)

(Received 13 April 1984)

For an "n-order" isothermal unimolecular decomposition using TG, we may write

$$d\alpha/dt = k(1-\alpha)^n \quad (1)$$

where n = reaction order, α = conversion, t = reaction time, and k = reaction constant. In the case of solid-state decompositions, only certain values of n are of theoretical significance. Thus, $n = 0$ is related to the Mampel power law (R1), $n = 1/2$ to contracting geometry involving cylindrical symmetry (R2), $n = 2/3$ to contracting geometry involving spherical symmetry (R3), and $n = 1$ to an Avrami–Erofeev mechanism (F1) [1]. In the present paper, a computer procedure is presented which involves numerical integration of eqn. (1) to obtain values of α. Thus, from $\alpha - t$ data pairs, the most appropriate from the preceding values of n can be determined ($n = 1/3$ is also included in the computer analysis and the program can be modified to include still other values of n) as well as approximate corresponding values of k.

THE COMPUTER PROGRAM

The computer program (CP) is listed in the Appendix. Briefly, the computer procedure involves the following. First, an approximate value of k is estimated by utilization of the expression, $k = \alpha/t$, which applies at low values of α. From this approximate value, the k-increment (S) can be estimated. Thus, from the initial values of the data in lines 280, 285, it can be ascertained that an increment of 0.001 should be adequate, cf. line 50 (for other data, it may be necessary to employ other S-values). Also, the initial value of k was taken as zero (lines 60, 110). Then, for particular values of n and k, values of α are calculated for the various values of t. These values of α are compared with corresponding data values and the absolute sum of their differences is obtained (MD) for each n and k, see lines 130–190 of CP. The

0040-6031/84/$03.00

value of k is then incremented and other values of MD are obtained. This process continues until the lowest value of MD is obtained. This value, MD(least), is stored and the next value of n is then selected, along with a value of k nearly equal to the last used value, cf. line 210 (a somewhat similar approach was used in a previous publication to estimate n and the activation energy for a non-isothermal reaction [2]). After various values of MD(least) have been obtained for the various n values employed, final values of n and k are determined based on the lowest value of MD(least), cf. lines 240–270.

In using the numerical integration procedure (lines 130–190), the following assumptions were made. The ordinates of the first interval under the curve [e.g., $(1-\alpha)^n$ vs. t] were assumed to be equal in order to estimate the initial average ordinate value (AV). From then on, AVs could be estimated (line 140) using the further assumption that the slopes of the curve were equal for adjacent intervals under the curve. Although smaller-sized intervals (DT) should afford more accurate values of α, it was found that whether such intervals of 0.1 or 1 were used, the final results were very similar. Similar remarks can be made with regard to the k-increment. There was little change in final results for the data in line 280 whether this increment was 0.0005 or 0.001. Thus, taking run time consideration into account, a value of DT = 1 was employed along with a k-increment of 0.001, which afforded a run time of ca. 4–5 min, using the seven pairs of $\alpha - t$ data in line 280. In the $\alpha - t$ data used, the time increments should be equal to the initial value of t, e.g., 30, 60, 90,...; or, 10, 20, 30,...; etc. Also, the number of data pairs (NN) should be adjusted in line 25, if necessary. Further, in the $\alpha - t$ data, values of α should not be too high (near unity) since for some reaction orders, $1-\alpha$, in the expression $(1-\alpha)^n$, may become negative thereby leading to an "illegal quantity error" statement.

In order to conserve space, only a portion of the computer printout of results is shown in the Appendix after the program listing for the data in line 280 [3]. Final values (based on an MD(least) of 0.015) of n and k were found to be 0.6667 and 0.022, respectively (theor., 2/3 and 0.022). For the data in line 285 (NN = 8) [4], final values of n and k were 1 and 0.017, respectively (theor., 1 and 0.0175), and the run time was ca. 4 min. From the preceding results, it can be seen that the type of reaction (n) can be predicted accurately, and the values obtained for k are in good agreement with expected values.

REFERENCES

1 T.P. Bagchi and P.K. Sen, Thermochim. Acta, 51 (1981) 175.
2 L. Reich and S.S. Stivala, Thermochim. Acta, 52 (1982) 337.
3 M.E. Brown and C.A.R. Phillpotts, J. Chem. Educ., 55 (1978) 556.
4 A.A. Frost and R.G. Pearson, Kinetics and Mechanism, Wiley, New York, 1953.

APPENDIX

A BASIC computer program and printout of results for the numerical analysis of isothermal TG data for mechanism (n) and reaction constant (k).

Program

```
10  REM *** NUMERICAL ANALYSIS TO SOLVE RATE EQN. FOR N AND K UNDER ISOTHE
    RMAL CONDITIONS ***
11 :
12  REM  *** N-VALUES USED: 0, 1/3, 1/2, 2/3, AND 1 ***
14 :
15  DIM MD(20),K(20),DM(20)
25 DT = 1:NN = 7:Z = 1:MD(0) = 100:DM(0) = 100
27 :
30  FOR L = 1 TO NN: READ A(L),T(L): NEXT L
35 :
40 Q = A(1) / T(1):R = 3 * Q
50 S = 0.001
60 Q = 0
70 :
75  IF Z > = 6 THEN 240
80 CC = 0:T = 0:AA = 0:AOLD = AA:DD = 0: ON Z GOSUB 330,340,350,360,370
100 :
110  FOR K = Q TO R STEP S
120 CC = CC + 1:DD = DD + 1
122 :
125  REM *** NUMERICAL INTEGRATION PROCEDURE ***
127 :
130  FOR I = 1 TO T(NN) / DT
140 AV = 1.5 * AA - .5 * AOLD
150 AOLD = AA
160 AA = AA + (K * (1 - AV) ^ N) * DT
170 T = T + DT
180  IF I * DT / T(1) =  INT (I * DT / T(1)) THEN  PRINT "T= "; INT (T * 1
    00 + .5) / 100;", ALPHA= "; INT (AA * 1E3 + .5) / 1E3;:MD(CC) = MD(CC
    ) +  ABS (A(DD) - AA): PRINT " , K= ";:ZZ = K: GOSUB 500:DD = DD + 1:
     PRINT
190  NEXT I
195 :
200  PRINT : PRINT "FOR N= "; INT (N * 100 + .5) / 100;", AND K= ";:ZZ = K
    : GOSUB 500: PRINT ", MD= ";:ZZ = MD(CC): GOSUB 500: PRINT : PRINT
205 :
210  IF MD(CC) > MD(CC - 1) THEN  PRINT "** MD(LEAST)= ";:ZZ = MD(CC - 1):
     GOSUB 500: PRINT " FOR K= ";:KK = (K - S):ZZ = KK: GOSUB 500: PRINT
    " AND N= ";N$:DM(Z) = MD(CC - 1):K(Z) = KK:N(Z) = N:Q = KK - S: PRINT
    :Z = Z + 1: GOSUB 400: GOTO 75
220  GOSUB 300
230  NEXT K
235 :
240  FOR J = 1 TO Z - 1
250  IF DM(J) > DM(J - 1) THEN DM(J) = DM(J - 1):K(J) = K(J - 1):N(J) = N(
    J - 1)
260  NEXT J
270  PRINT : PRINT "*** FINAL VALUES: MD= ";:ZZ = DM(J - 1): GOSUB 500: PRINT
    ", K= ";:ZZ = K(J - 1): GOSUB 500: PRINT ", AND N= ";:ZZ = N(J - 1): GOSUB
    500
275  END
277 :
280  DATA   .203,10,.377,20,.523,30,.645,40,.743,50,.822,60,.883,70: REM
          ** THEOR. ISOTH. DATA, BROWN, J.CHEM.ED.,55,556,'84 **
285  REM DATA .153,10,.283,20,.4,30,.5,40,.58,50,.65,60,.71,70,.755,80: RE
    M  ** ISOTH. EXPTL. DATA, FROST/PEARSON, P.50 **
290 :
300  REM  **SUBRTN. TO REINITIALIZE VALUES**
305 :
310 AA = 0:AOLD = AA:DD = 0:T = 0
320  RETURN
325 :
330  REM  **SUBRTN. TO USE DIFFERENT N-VALUES**
```

```
333 :
335 N = 0:N$ = "0": RETURN
340 N = 1 / 3:N$ = "1/3": RETURN
350 N = 1 / 2:N$ = "1/2": RETURN
360 N = 2 / 3:N$ = "2/3": RETURN
370 N = 1:N$ = "1": RETURN
390 :
400  REM  **SUBRTN. TO CLEAR MD'S**
405 :
410  FOR M = 1 TO CC:MD(M) = 0: NEXT M
420  RETURN
430 :
500  REM  **SUBRTN. FOR DEC. PLACES**
505 :
510 ZZ =  INT (ZZ * 1E4 + .5) / 1E4: PRINT ZZ;
520  RETURN
```

Results

```
T= 10, ALPHA= .178 , K= .019
T= 20, ALPHA= .334 , K= .019
T= 30, ALPHA= .469 , K= .019
T= 40, ALPHA= .584 , K= .019
T= 50, ALPHA= .681 , K= .019
T= 60, ALPHA= .762 , K= .019
T= 70, ALPHA= .827 , K= .019

FOR N= .67, AND K= .019, MD= .3612

T= 10, ALPHA= .187 , K= .02
T= 20, ALPHA= .349 , K= .02
T= 30, ALPHA= .488 , K= .02
T= 40, ALPHA= .606 , K= .02
T= 50, ALPHA= .704 , K= .02
T= 60, ALPHA= .784 , K= .02
T= 70, ALPHA= .848 , K= .02

FOR N= .67, AND K= .02, MD= .2301

T= 10, ALPHA= .196 , K= .021
T= 20, ALPHA= .364 , K= .021
T= 30, ALPHA= .507 , K= .021
T= 40, ALPHA= .627 , K= .021
T= 50, ALPHA= .725 , K= .021
T= 60, ALPHA= .805 , K= .021
T= 70, ALPHA= .867 , K= .021

FOR N= .67, AND K= .021, MD= .1048

T= 10, ALPHA= .204 , K= .022
T= 20, ALPHA= .379 , K= .022
T= 30, ALPHA= .526 , K= .022
T= 40, ALPHA= .647 , K= .022
T= 50, ALPHA= .746 , K= .022
T= 60, ALPHA= .824 , K= .022
T= 70, ALPHA= .885 , K= .022

FOR N= .67, AND K= .022, MD= .0148

T= 10, ALPHA= .213 , K= .023
T= 20, ALPHA= .393 , K= .023
T= 30, ALPHA= .544 , K= .023
T= 40, ALPHA= .667 , K= .023
T= 50, ALPHA= .765 , K= .023
T= 60, ALPHA= .843 , K= .023
T= 70, ALPHA= .901 , K= .023

FOR N= .67, AND K= .023, MD= .1289

** MD(LEAST)= .0148 FOR K= .022 AND N= 2/3
```

```
T= 10, ALPHA= .19 , K= .021
T= 20, ALPHA= .343 , K= .021
T= 30, ALPHA= .467 , K= .021
T= 40, ALPHA= .568 , K= .021
T= 50, ALPHA= .65 , K= .021
T= 60, ALPHA= .716 , K= .021
T= 70, ALPHA= .77 , K= .021

FOR N= 1, AND K= .021, MD= .4911

T= 10, ALPHA= .198 , K= .022
T= 20, ALPHA= .356 , K= .022
T= 30, ALPHA= .483 , K= .022
T= 40, ALPHA= .585 , K= .022
T= 50, ALPHA= .667 , K= .022
T= 60, ALPHA= .733 , K= .022
T= 70, ALPHA= .786 , K= .022

FOR N= 1, AND K= .022, MD= .3882

T= 10, ALPHA= .206 , K= .023
T= 20, ALPHA= .369 , K= .023
T= 30, ALPHA= .498 , K= .023
T= 40, ALPHA= .602 , K= .023
T= 50, ALPHA= .683 , K= .023
T= 60, ALPHA= .748 , K= .023
T= 70, ALPHA= .8 , K= .023

FOR N= 1, AND K= .023, MD= .295

T= 10, ALPHA= .214 , K= .024
T= 20, ALPHA= .381 , K= .024
T= 30, ALPHA= .513 , K= .024
T= 40, ALPHA= .617 , K= .024
T= 50, ALPHA= .699 , K= .024
T= 60, ALPHA= .763 , K= .024
T= 70, ALPHA= .814 , K= .024

FOR N= 1, AND K= .024, MD= .225

T= 10, ALPHA= .221 , K= .025
T= 20, ALPHA= .394 , K= .025
T= 30, ALPHA= .528 , K= .025
T= 40, ALPHA= .632 , K= .025
T= 50, ALPHA= .713 , K= .025
T= 60, ALPHA= .777 , K= .025
T= 70, ALPHA= .826 , K= .025

FOR N= 1, AND K= .025, MD= .184

T= 10, ALPHA= .229 , K= .026
T= 20, ALPHA= .406 , K= .026
T= 30, ALPHA= .542 , K= .026
T= 40, ALPHA= .647 , K= .026
T= 50, ALPHA= .727 , K= .026
T= 60, ALPHA= .79 , K= .026
T= 70, ALPHA= .838 , K= .026

FOR N= 1, AND K= .026, MD= .1677

T= 10, ALPHA= .237 , K= .027
T= 20, ALPHA= .417 , K= .027
T= 30, ALPHA= .555 , K= .027
T= 40, ALPHA= .66 , K= .027
T= 50, ALPHA= .741 , K= .027
T= 60, ALPHA= .802 , K= .027
T= 70, ALPHA= .849 , K= .027

FOR N= 1, AND K= .027, MD= .1781

** MD(LEAST)= .1677 FOR K= .026 AND N= 1


*** FINAL VALUES: MD= .0148, K= .022, AND N= .6667
]
```